Lesson 24

HOW TO USE GAP (GROUPS, ALGORTHIMS, AND PROGRAMS)

1. How can I redisplay the previous command in order to edit it?

   Press down on the control key and then also press p. In other words, "Ctrl p".

2. If the program gets in a loop and shows you the prompt "brk>" instead of "gap>", how can I exit the loop?

   Press down on the control key and then also press d. In other words, "Ctrl d".

3. How can I exit the program?

   Either click on the "close" box for the window, or type "quit;" and press "Enter."

4. How do I find the inverse of a permutation?

```
gap> a:=(1,2,3,4);
(1,2,3,4)

gap> a^-1;
(1,4,3,2)
```

5. How do I find the conjugate of a permutation in the form $a^b = b^{-1}ab$?

```
gap> a:=(1,2,3,4,5);
(1,2,3,4,5)

gap> b:=(2,4,5);
(2,4,5)

gap> a^b;
(1,4,3,5,2)

gap> b^-1*a*b;
(1,4,3,5,2)
```

6. How can I create a group from permutations, find the size of the group, and find the elements in the group?

```
gap> a:=(1,2);
(1,2)

gap> b:=(1,2,3);
(1,2,3)

gap> s3:=Group(a,b);
Group([ (1,2), (1,2,3) ])

gap> Size(s3);
6

gap> Elements(s3);
[ (), (2,3), (1,2), (1,2,3), (1,3,2), (1,3) ]

gap> d3:=Group([(1,2),(1,2,3)]);
Group([ (1,2), (1,2,3) ])
```

7. How do I create a symmetric group of degree n with n! elements?

```
gap> s4:=SymmetricGroup(4);
Sym( [ 1 .. 4 ] )

gap> Elements(s4);
[ (), (3,4), (2,3), (2,3,4), (2,4,3), (2,4), (1,2), (1,2)(3,4), (1,2,3),
(1,2,3,4), (1,2,4,3), (1,2,4), (1,3,2),
  (1,3,4,2), (1,3), (1,3,4), (1,3)(2,4), (1,3,2,4), (1,4,3,2), (1,4,2),
(1,4,3), (1,4), (1,4,2,3), (1,4)(2,3) ]
```

8. How do I create a dihedral group for an n-sided regular polygon with 2n elements?

```
gap> d4:=DihedralGroup(IsPermGroup,8);
Group([ (1,2,3,4), (2,4) ])

gap> Elements(d4);
[ (), (2,4), (1,2)(3,4), (1,2,3,4), (1,3), (1,3)(2,4), (1,4,3,2),
(1,4)(2,3) ]
```

9. How do I find *Q*, the quaternion group?

```
gap> q:=QuaternionGroup(IsPermGroup,8);
Group([ (1,5,3,7)(2,8,4,6), (1,2,3,4)(5,6,7,8) ])

gap> Elements(q);
[ (), (1,2,3,4)(5,6,7,8), (1,3)(2,4)(5,7)(6,8), (1,4,3,2)(5,8,7,6),
(1,5,3,7)(2,8,4,6), (1,6,3,8)(2,5,4,7),
  (1,7,3,5)(2,6,4,8), (1,8,3,6)(2,7,4,5) ]
```

10. How do I find the multiplication table for a group?

```
gap> d3:=DihedralGroup(IsPermGroup,6);
Group([ (1,2,3), (2,3) ])

gap> ShowMultiplicationTable(d3);


*       | ()      (2,3)   (1,2)   (1,2,3) (1,3,2) (1,3)
--------+-----------------------------------------------
()      | ()      (2,3)   (1,2)   (1,2,3) (1,3,2) (1,3)
(2,3)   | (2,3)   ()      (1,2,3) (1,2)   (1,3)   (1,3,2)
(1,2)   | (1,2)   (1,3,2) ()      (1,3)   (2,3)   (1,2,3)
(1,2,3) | (1,2,3) (1,3)   (2,3)   (1,3,2) ()      (1,2)
(1,3,2) | (1,3,2) (1,2)   (1,3)   ()      (1,2,3) (2,3)
(1,3)   | (1,3)   (1,2,3) (1,3,2) (2,3)   (1,2)   ()
```

11. How do I tell if a group is abelian?

```
gap> d3:=DihedralGroup(IsPermGroup,6);
Group([ (1,2,3), (2,3) ])

gap> IsAbelian(d3);
false
```

12. How do I determine if a group is simple? (A group is simple if its only normal subgroups are itself and the identity.)

```
gap> d4:=DihedralGroup(IsPermGroup,8);
Group([ (1,2,3,4), (2,4) ])

gap> IsSimple(d4);
false
```

13. How do I find the conjugacy classes for a group?

```
gap> d4:=DihedralGroup(IsPermGroup,8);
Group([ (1,2,3,4), (2,4) ])

gap> Elements(d4);
[ (), (2,4), (1,2)(3,4), (1,2,3,4), (1,3), (1,3)(2,4), (1,4,3,2),
(1,4)(2,3) ]

gap> c:=ConjugacyClasses(d4);
[ ()^G, (2,4)^G, (1,2)(3,4)^G, (1,2,3,4)^G, (1,3)(2,4)^G ]

gap> Elements(c[1]);
[ () ]
gap> Elements(c[2]);
[ (2,4), (1,3) ]

gap> Elements(c[3]);
[ (1,2)(3,4), (1,4)(2,3) ]

gap> Elements(c[4]);
[ (1,2,3,4), (1,4,3,2) ]

gap> Elements(c[5]);
[ (1,3)(2,4) ]

gap> List(c,i->Elements(i));
[ [ () ], [ (2,4), (1,3) ], [ (1,2)(3,4), (1,4)(2,3) ], [ (1,2,3,4),
(1,4,3,2) ], [ (1,3)(2,4) ] ]
```

14. How do I find the orbit of a point that a group acts on?

```
gap> s3:=SymmetricGroup(3);
Sym( [ 1 .. 3 ] )

gap> Orbit(s3,1);
[ 1, 3, 2 ]
```

15. How do I find all of the orbits for a set of points that a group acts on?

```
gap> g:=Group([(1,2),(3,4)]);
Group([ (1,2), (3,4) ])

gap> Elements(g);
[ (), (3,4), (1,2), (1,2)(3,4) ]

gap> Orbits(g);
[ [ 1, 2 ], [ 3, 4 ] ]
```

16. How do I find all subgroups of a group?

```
gap> d3:=DihedralGroup(IsPermGroup,6);
Group([ (1,2,3), (2,3) ])

gap> h:=AllSubgroups(d3);
[ Group(()), Group([ (2,3) ]), Group([ (1,2) ]), Group([ (1,3) ]),
Group([ (1,3,2) ]), Group([ (1,3,2), (2,3) ]) ]

gap> List(h,i->Elements(i));
[ [ () ], [ (), (2,3) ], [ (), (1,2) ], [ (), (1,3) ], [ (), (1,2,3),
(1,3,2) ], [ (), (2,3), (1,2), (1,2,3), (1,3,2), (1,3) ] ]

gap> d4:=DihedralGroup(IsPermGroup,8);
Group([ (1,2,3,4), (2,4) ])

gap> h:=AllSubgroups(d4);
[ Group(()), Group([ (1,3)(2,4) ]), Group([ (2,4) ]), Group([ (1,3) ]),
Group([ (1,2)(3,4) ]), Group([ (1,4)(2,3) ]),
  Group([ (1,3)(2,4), (2,4) ]), Group([ (1,3)(2,4), (1,2,3,4) ]),
Group([ (1,3)(2,4), (1,2)(3,4) ]), Group([ (1,3)
  (2,4), (2,4), (1,2,3,4) ]) ]

gap> Elements(h[1]);
[ () ]

gap> Elements(h[2]);
[ (), (1,3)(2,4) ]

gap> Elements(h[3]);
[ (), (2,4) ]

gap> Elements(h[4]);
[ (), (1,3) ]

gap> Elements(h[5]);
[ (), (1,2)(3,4) ]

gap> Elements(h[6]);
[ (), (1,4)(2,3) ]

gap> Elements(h[7]);
[ (), (2,4), (1,3), (1,3)(2,4) ]

gap> Elements(h[8]);
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2) ]

gap> Elements(h[9]);
[ (), (1,2)(3,4), (1,3)(2,4), (1,4)(2,3) ]

gap> Elements(h[10]);
[ (), (2,4), (1,2)(3,4), (1,2,3,4), (1,3), (1,3)(2,4), (1,4,3,2),
(1,4)(2,3) ]
```

17. How do I find the subgroup generated by particular permutations?

```
gap> d3:=DihedralGroup(IsPermGroup,6);
Group([ (1,2,3), (2,3) ])

gap> h2:=Subgroup(d3,[(1,2,3)]);
Group([ (1,2,3) ])

gap> Elements(h2);
[ (), (1,2,3), (1,3,2) ]
```

18. How do I determine if a subgroup is normal?

```
gap> d4:=DihedralGroup(IsPermGroup,8);
Group([ (1,2,3,4), (2,4) ])

gap> Elements(d4);
[ (), (2,4), (1,2)(3,4), (1,2,3,4), (1,3), (1,3)(2,4), (1,4,3,2),
(1,4)(2,3) ]

gap> h:=Subgroup(d4,[(1,2,3,4)]);
Group([ (1,2,3,4) ])

gap> Elements(h);
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2) ]

gap> IsNormal(d4,h);
true
```

19. How do I find the center of a group?

```
gap> d4:=DihedralGroup(IsPermGroup,8);
Group([ (1,2,3,4), (2,4) ])

gap> c:=Center(d4);
Group([ (1,3)(2,4) ])

gap> Elements(c);
[ (), (1,3)(2,4) ]
```

20. How do I find the derived subgroup (i.e.commutator subgroup) of a group?

```
gap> d4:=DihedralGroup(IsPermGroup,8);
Group([ (1,2,3,4), (2,4) ])

gap> ds:=DerivedSubgroup(d4);
Group([ (1,3)(2,4) ])

gap> Elements(ds);
[ (), (1,3)(2,4) ]
```

21. How do I find the alternating subgroup of degree n (with n!/2 elements)?

```
gap> alt:=AlternatingGroup(3);
Alt( [ 1 .. 3 ] )

gap> Elements(alt);
[ (), (1,2,3), (1,3,2) ]
```

22. How do I find a Sylow Subgroup?

```
gap> s3:=SymmetricGroup(3);
Sym( [ 1 .. 3 ] )

gap> sylow3:=SylowSubgroup(s3,3);
Group([ (1,2,3) ])

gap> Elements(sylow3);
[ (), (1,2,3), (1,3,2) ]
```

23. How do I find all Sylow subgroups of a group for a given prime?

```
gap> s4:=SymmetricGroup(4);
Sym( [ 1 .. 4 ] )

gap> Size(s4);
24

gap> h:=SylowSubgroup(s4,3);
Group([ (1,2,3) ])

gap> Elements(h);
[ (), (1,2,3), (1,3,2) ]

gap> w:=ConjugateSubgroups(s4,h);
[ Group([ (1,2,3) ]), Group([ (2,3,4) ]), Group([ (1,3,4) ]),
Group([ (1,4,2) ]) ]

gap> List(w,i->Elements(i));
[ [ (), (1,2,3), (1,3,2) ], [ (), (2,3,4), (2,4,3) ], [ (), (1,3,4),
(1,4,3) ], [ (), (1,2,4), (1,4,2) ] ]

gap> Elements(w[1]);
[ (), (1,2,3), (1,3,2) ]

gap> Elements(w[2]);
[ (), (2,3,4), (2,4,3) ]

gap> Elements(w[3]);
[ (), (1,3,4), (1,4,3) ]

gap> Elements(w[4]);
[ (), (1,2,4), (1,4,2) ]
```

24. How do I find the stabilizer group for one point or for several points of a set that a group acts on?

```
gap> a:=(1,2);
(1,2)

gap> b:=(3,4);
(3,4)

gap> c:=(5,6);
(5,6)

gap> g:=Group(a,b,c);
Group([ (1,2), (3,4), (5,6) ])

gap> Size(g);
8

gap> Elements(g);
[ (), (5,6), (3,4), (3,4)(5,6), (1,2), (1,2)(5,6), (1,2)(3,4),
(1,2)(3,4)(5,6) ]

gap> h:=Stabilizer(g,1);
Group([ (5,6), (3,4) ])

gap> Elements(h);
[ (), (5,6), (3,4), (3,4)(5,6) ]

gap> h2:=Stabilizer(g,[1,3],OnTuples);
Group([ (5,6) ])

gap> Elements(h2);
[ (), (5,6) ]
```

25. How do I find the right cosets of a subset *H* of *G*?

```
gap> g:=Group([(1,2,3),(1,2)]);
Group([ (1,2,3), (1,2) ])

gap> Elements(g);
[ (), (2,3), (1,2), (1,2,3), (1,3,2), (1,3) ]

gap> h:=Subgroup(g,[(1,2)]);
Group([ (1,2) ])

gap> Elements(h);
[ (), (1,2) ]

gap> c:=RightCosets(g,h);
[ RightCoset(Group( [ (1,2) ] ),()), RightCoset(Group( [ (1,2) ] ),(1,3,2)),
RightCoset(Group( [ (1,2) ] ),(1,2,3)) ]

gap> List(c,i->Elements(i));
[ [ (), (1,2) ], [ (2,3), (1,3,2) ], [ (1,2,3), (1,3) ] ]

gap> Elements(c[1]);
[ (), (1,2) ]

gap> Elements(c[2]);
[ (2,3), (1,3,2) ]

gap> Elements(c[3]);
[ (1,2,3), (1,3) ]
```